

```

*title: Example data on the two hospitals
* Example data generated by -dataex-. To install: ssc install dataex

*** go to correct working path/folder

* cd "z:\Zwahlen_1\pubprojects\2016_17_evidence_based_mental_health\data_example"

* Input the data set
clear
input byte risk long(sex age) float relapse double hospital int N
1 1 0 0 0 198
1 1 0 1 0 2
1 1 0 0 1 99
1 1 0 1 1 1
2 1 0 0 0 57
2 1 0 1 0 3
2 1 0 0 1 38
2 1 0 1 1 2
1 1 1 0 0 96
1 1 1 1 0 4
1 1 1 0 1 48
1 1 1 1 1 2
2 1 1 0 0 34
2 1 1 1 0 6
2 1 1 0 1 85
2 1 1 1 1 15
1 2 0 0 0 196
1 2 0 1 0 4
1 2 0 0 1 49
1 2 0 1 1 1
2 2 0 0 0 54
2 2 0 1 0 6
2 2 0 0 1 90
2 2 0 1 1 10
1 2 1 0 0 95
1 2 1 1 0 5
1 2 1 0 1 190
1 2 1 1 1 10
2 2 1 0 0 30
2 2 1 1 0 10
2 2 1 0 1 120
2 2 1 1 1 40
end
label values risk rp_1
label def rp_1 1 "low", modify
label def rp_1 2 "high", modify
label values sex sex_1
label def sex_1 1 "female", modify
label def sex_1 2 "male", modify
label values age age_1
label def age_1 0 "<60", modify
label def age_1 1 "60+", modify
label values relapse relapse
label def relapse 0 "no relapse", modify
label def relapse 1 "relapse", modify
label values hospital hosp
label def hosp 0 "A", modify
label def hosp 1 "B", modify
label var relapse "6-month relapse"
expand N
drop N

* data set is now ready: describe now data and labelling
codebook

```

```

tabulate hospital relapse, row

*** IPTW
*** model for being in which hospital including all interactions

logistic hospital i.sex##i.age##i.risk

cap drop p
predict p

tab p
summ p
cap drop iptw
generate iptw =1/p if hospital==1
replace iptw =1/(1-p) if hospital==0

* check that now risk factors are balanced...

foreach var of varlist sex age risk {
    tabulate `var' hospital, col
    tabulate `var' hospital [iweight=iptw], col
}

* expected relapse rate for both hospitals

tabulate hospital relapse [iweight=iptw], row

* to obtain proper 95% CI with robust standard erros for risk per hospital one needs to
svyset the data

svyset [pweight= iptw]
svy : tabulate  hospital relapse  , row percent ci format(%5.2f)

*** IPTW analysis with robust standard erros

* causal risk difference
binreg relapse hospital [pweight=iptw], vce(robust) rd
* causal risk ratio
binreg relapse hospital [pweight=iptw], vce(robust) rr
* causal odds ratio
binreg relapse hospital [pweight=iptw], vce(robust) or
* causal odds ratio
logistic relapse hospital [pweight=iptw], vce(robust)

* generate a table with the weights by age sex severity

preserve

** check data by collapsing data set

collapse (mean) iptw , by(sex age risk hospital)
gsort hosp -sex age risk
bysort hosp : list sex age risk iptw, clean noobs

restore

```

```
/*
```

sex	age	risk	iptw
male	<60	low	1.25
male	<60	high	2.67
male	60+	low	3.0
male	60+	high	5.0
female	<60	low	1.5
female	<60	high	1.67
female	60+	low	1.5
female	60+	high	3.5

```
-----  
-----  
-> hospital = B
```

sex	age	risk	iptw
male	<60	low	5.0
male	<60	high	1.6
male	60+	low	1.5
male	60+	high	1.25
female	<60	low	3.0
female	<60	high	2.5
female	60+	low	3.0
female	60+	high	1.4

```
*/
```

```
*iptw weights without interactions
```

```
logistic hospital i.sex i.age i.risk
```

```
cap drop p
```

```
predict p
```

```
tab p
```

```
summ p
```

```
cap drop iptw
```

```
generate iptw =1/p if hospital==1
```

```
replace iptw =1/(1-p) if hospital==0
```

```
* causal odds ratio
```

```
logistic relapse hospital [pweight=iptw], vce(robust)
```

```
* naive estimates
```

```
logistic relapse hospital
```

```
logistic relapse hospital sex age risk
```

```
logistic relapse hospital i.sex##i.age##i.risk
```